

## Linguagem LALG

1. <programa> ::= program ident ; <corpo> .
2. <corpo> ::= <dc> begin <comandos> end
3. <dc> ::= <dc\_v> <dc\_p>
4. <dc\_v> ::= var <variaveis> : <tipo\_var> ; <dc\_v> | λ
5. <tipo\_var> ::= real | integer
6. <variaveis> ::= ident <mais\_var>
7. <mais\_var> ::= , <variaveis> | λ
8. <dc\_p> ::= procedure ident <parametros> ; <corpo\_p> <dc\_p> | λ
9. <parametros> ::= ( <lista\_par> ) | λ
10. <lista\_par> ::= <variaveis> : <tipo\_var> <mais\_par>
11. <mais\_par> ::= ; <lista\_par> | λ
12. <corpo\_p> ::= <dc\_loc> begin <comandos> end ;
13. <dc\_loc> ::= <dc\_v>
14. <lista\_arg> ::= ( <argumentos> ) | λ
15. <argumentos> ::= ident <mais\_ident>
16. <mais\_ident> ::= ; <argumentos> | λ
17. <pfalsa> ::= else <cmd> | λ
18. <comandos> ::= <cmd> ; <comandos> | λ
19. <cmd> ::= read ( <variaveis> ) |  
                  write ( <variaveis> ) |  
                  while <condicao> do <cmd> |  
                  if <condicao> then <cmd> <pfalsa> |  
                  ident := <expressao> |  
                  ident <lista\_arg> |  
                  begin <comandos> end
20. <condicao> ::= <expressao> <relacao> <expressao>
21. <relacao> ::= = | <> | >= | <= | > | <
22. <expressao> ::= <termo> <outros\_termos>
23. <op\_un> ::= + | - | λ
24. <outros\_termos> ::= <op\_ad> <termo> <outros\_termos> | λ
25. <op\_ad> ::= + | -
26. <termo> ::= <op\_un> <fator> <mais\_fatores>
27. <mais\_fatores> ::= <op\_mul> <fator> <mais\_fatores> | λ
28. <op\_mul> ::= \* | /
29. <fator> ::= ident | numero\_int | numero\_real | ( <expressao> )

Além disso, inclusão do comando repeat-until.

Comentários entre chaves { }

Identificadores e números são itens léxicos da forma:

- ident: seqüência de letras e dígitos, começando por letra
- número inteiro: seqüência de dígitos (0 a 9)
- número real: pelo menos um dígito seguido de um ponto decimal e de uma seqüência de um ou mais dígitos

### Exemplos de programas LALG

```
program nome1;  
{exemplo 1}  
var a, a1, b: integer;  
begin  
read(a, b);  
a1:= a + b;  
while a1>a do  
begin  
write(a1);  
a1:= a1-1;  
end;  
if a<> b then write(a);  
end.  
  
program nome2;  
{exemplo 2}  
var a: real;  
var b: integer;  
procedure nomep(x: real);  
var a, c: integer;  
begin  
read(c, a);  
if a<x+c then  
begin  
a:= c+x;  
write(a);  
end  
else c:= a+x;  
end;  
begin {programa principal}  
read(b);  
nomep(b);  
end.
```